Pensieve header: A free-Lie calculator.

A Lyndon word is a word lexicographically smaller than all of its proper right factors; see http://katlas.math.toronto.edu/drorbn/bbs/show?shot=Chu-071214-182203.jpg

```
LyndonQ[w_String] := And @@ (
    OrderedQ[{w, #}] & /@ Table[StringDrop[w, i], {i, 1, StringLength[w] - 1}]
   );
AllWords[0, _List] = {""};
AllWords[n_ /; n > 0, ab_List] := AllWords[n, ab] = Flatten[Outer[
     StringJoin[#1, #2] &,
     AllWords[n - 1, ab],
     ab
    ]];
AllLyndonWords[n_Integer, ab_List] := LW /@ Select[AllWords[n, ab], LyndonQ];
AllLyndonWords[{n_}, ab_List] := Join @@ Table[AllLyndonWords[k, ab], {k, n}];
LyndonFactorization[LW[w_String] /; StringLength[w] == 1] := LW[w];
LyndonFactorization[LW[w_String] /; StringLength[w] > 1] := Module[
   {rf},
   rf = First[Sort[Table[StringDrop[w, i], {i, 1, StringLength[w] - 1}]]];
   LW /@ {StringDrop[w, -StringLength[rf]], rf}
  ];
LW /: LW[x_] ≤ LW[y_] := OrderedQ[{x, y}];
LW /: x_LW ≥ y_LW := y ≤ x;
LW /: x_LW > y_LW := ! (x ≤ y);
LW /: x_LW < y_LW := ! (y ≤ x);
Format[LW[w_], StandardForm] := Defer[⟨w⟩];
⟨w_⟩ := LW[w];
Deg[LW[x_]] := StringLength[x];
```

```
{LyndonQ["abba"], LyndonQ["ababb"]}
```

```
{False, True}
```

```
{AllWords[3, {"1", "2"}], AllLyndonWords[{3}, {"1", "2"}]}
```

```
{{111, 112, 121, 122, 211, 212, 221, 222}, {⟨1⟩, ⟨2⟩, ⟨12⟩, ⟨112⟩, ⟨122⟩}}
```

```
Table[Length[AllLyndonWords[k, {"1", "2"}]], {k, 10}]
```

```
{2, 1, 2, 3, 6, 9, 18, 30, 56, 99}
```

```
Table[Length[AllLyndonWords[k, {"1", "2", "3"}]], {k, 8}]
```

```
{3, 3, 8, 18, 48, 116, 312, 810}
```

```
TreeForm[LW["12122"] //. w_LW ⧴ LyndonFactorization[w] /. LW[w_] ⧴ w]
```




```
b[0, _] = 0; b[_, 0] = 0;
b[c_ * x_LW, y_] := Expand[c b[x, y]];
b[x_, c_ * y_LW] := Expand[c b[x, y]];
b[x_Plus, y_] := b[#, y] & /@ x;
b[x_, y_Plus] := b[x, #] & /@ y;
b[w_LW, z_LW] := Which[
   w === z, 0,
   z < w, Expand[-b[z, w]],
   Deg[w] == 1, LW[First[w] <> First[z]],
   True, Module[{x, y},
    {x, y} = LyndonFactorization[w];
    If[y ≥ z,
     LW[First[w] <> First[z]],
     b[w, z] = b[x, b[y, z]] + b[b[x, z], y]
    ]
   ]
  ];
```

```
b[LW["112"], LW["122"]]
```

⟨112122⟩ + ⟨112212⟩

```
Outer[b, AllLyndonWords[{3}, {"1", "2"}],
  AllLyndonWords[{3}, {"1", "2"}]] // MatrixForm
```

0	⟨12⟩	⟨112⟩	⟨1112⟩	⟨1122⟩
−⟨12⟩	0	−⟨122⟩	−⟨1122⟩	−⟨1222⟩
−⟨112⟩	⟨122⟩	0	−⟨11212⟩	⟨12122⟩
−⟨1112⟩	⟨1122⟩	⟨11212⟩	0	⟨112122⟩ + ⟨112212⟩
−⟨1122⟩	⟨1222⟩	−⟨12122⟩	−⟨112122⟩ − ⟨112212⟩	0

```
Union[Flatten[Outer[(b[#1, #2] + b[#2, #1]) &,
   AllLyndonWords[{6}, {"1", "2"}], AllLyndonWords[{6}, {"1", "2"}]
  ]]]
```

```
{0}
```

```
Outer[(b[#1, b[#2, #3]] + b[#2, b[#3, #1]] + b[#3, b[#1, #2]]) &,
   AllLyndonWords[{5}, {"1", "2"}],
   AllLyndonWords[{5}, {"1", "2"}], AllLyndonWords[{5}, {"1", "2"}]
  ] // Flatten //
 Union
```

```
{0}
```

```
{xw = LW["x"], yw = LW["y"]}
```

```
{⟨x⟩, ⟨y⟩}
```

```
ad[x_][y_] := b[x, y];
MakeLieSeries[d_, l_List] := MakeLieSeries[d, #] & /@ l;
MakeLieSeries[d_, s_LieSeries] /; Length[s] ≤ d := s;
MakeLieSeries[d_, s_LieSeries] /; Length[s] > d := Take[s, d];
MakeLieSeries[d_, a_ → b_] := (a → MakeLieSeries[d, b]);
MakeLieSeries[d_, expr_] := LieSeries @@ Table[
     expr /. w_LW /; Deg[w] ≠ k → 0,
     {k, d}
    ];
LieSeries /: s1_LieSeries + s2_LieSeries := Module[
   {l = Min[Length /@ {s1, s2}]},
   LieSeries @@ (Take[List @@ s1, l] + Take[List @@ s2, l])
  ];
LieSeries /: c_ * s_LieSeries := Expand[c * #] & /@ s;
b[s1_LieSeries, s2_LieSeries] := Module[
   {d, k, m1, m2},
   m1 = 1 + LengthWhile[s1, # == 0 &];
   m2 = 1 + LengthWhile[s2, # == 0 &];
   LieSeries @@ Table[
     Sum[b[s1[[k]], s2[[d - k]]], {k, m1, d - m2}],
     {d, Min[m1 + Length[s2], m2 + Length[s1]]}
    ]
  ];
LieSeries /: EulerE[s_LieSeries] :=
  LieSeries @@ Expand[Range[Length[s]] * (List @@ s)];
OperatorSeries[f_, var_ → op_, d_][ψ_] := Module[
  {ser, as, ni, nf, t, l},
  ser = List @@ Series[f, {var, 0, d}];
  {as, ni, nf} = ser[[{3, 4, 5}]];
  t = Nest[op, ψ, ni];
  If[Head[t] === LieSeries,
   l = Length[t];
   Expand[as.NestList[MakeLieSeries[l, op[#]] &, t, nf - ni - 1]],
   Expand[as.NestList[op, t, nf - ni - 1]]
  ]
 ]
```

```
OperatorSeries[E^(-ad), ad → ad[yw], 3][xw]
```

$$\langle x\rangle + \langle xy\rangle + \frac{\langle xyy\rangle}{2} + \frac{\langle xyyy\rangle}{6}$$

```
MakeLieSeries[5, OperatorSeries[E^(-ad), ad → ad[xw], 3][yw]]
```

$$\text{LieSeries}\left[\langle y\rangle,\ -\langle xy\rangle,\ \frac{\langle xxy\rangle}{2},\ -\frac{\langle xxxy\rangle}{6},\ 0\right]$$

```
{
 t1 = MakeLieSeries[2, ⟨"y"⟩ + OperatorSeries[E^(-ad), ad → ad[⟨"y"⟩], 1][⟨"x"⟩]],
 bch = Append[LieSeries[⟨"x"⟩ + ⟨"y"⟩], 0];
 t2 = OperatorSeries[(1 - E^(-ad)) / ad, ad → ad[bch], 1][EulerE[bch]],
 t1 - t2
}
```

{LieSeries[⟨x⟩ + ⟨y⟩, ⟨xy⟩], LieSeries[⟨x⟩ + ⟨y⟩, 0], LieSeries[0, ⟨xy⟩]}

```
BCH[1] = LieSeries[⟨"x"⟩ + ⟨"y"⟩];
BCH[n_] := BCH[n] = Module[
   {bch, t1, t2},
   bch = Append[BCH[n - 1], 0];
   t1 =
    MakeLieSeries[n, ⟨"y"⟩ + OperatorSeries[E^(-ad), ad → ad[⟨"y"⟩], n - 1][⟨"x"⟩]];
   t2 = OperatorSeries[(1 - E^(-ad)) / ad, ad → ad[bch], n - 1][EulerE[bch]];
   bch + (t1 - t2) / n
  ]
```

```
BCH[2]
```

$$\text{LieSeries}\left[\langle x\rangle + \langle y\rangle,\ \frac{\langle xy\rangle}{2}\right]$$

```
BCH[8]
```

$$\begin{aligned}
&\text{LieSeries}\Big[\langle x\rangle + \langle y\rangle,\ \frac{\langle xy\rangle}{2},\ \frac{\langle xxy\rangle}{12} + \frac{\langle xyy\rangle}{12},\ \frac{\langle xxyy\rangle}{24},\\
&-\frac{\langle xxxxy\rangle}{720} + \frac{\langle xxxyy\rangle}{180} + \frac{\langle xxyxy\rangle}{360} + \frac{\langle xxyyy\rangle}{180} + \frac{\langle xyxyy\rangle}{120} - \frac{\langle xyyyy\rangle}{720},\\
&-\frac{\langle xxxxyy\rangle}{1440} + \frac{\langle xxxyxy\rangle}{720} + \frac{\langle xxxyyy\rangle}{360} + \frac{\langle xxyxyy\rangle}{240} - \frac{\langle xxyyyy\rangle}{1440},\\
&\frac{\langle xxxxxxy\rangle}{30\,240} - \frac{\langle xxxxxyy\rangle}{5040} + \frac{\langle xxxxyxy\rangle}{10\,080} + \frac{\langle xxxxyyy\rangle}{3780} + \frac{\langle xxxyxxy\rangle}{10\,080} + \frac{\langle xxxyxyy\rangle}{1680} +\\
&\frac{\langle xxxyyxy\rangle}{1260} + \frac{\langle xxxyyyy\rangle}{3780} + \frac{\langle xxyxxyy\rangle}{2016} - \frac{\langle xxyxyxy\rangle}{5040} + \frac{13\,\langle xxyxyyy\rangle}{15\,120} + \frac{\langle xxyyxyy\rangle}{10\,080} -\\
&\frac{\langle xxyyyxy\rangle}{1512} - \frac{\langle xxyyyyy\rangle}{5040} + \frac{\langle xyxyxyy\rangle}{1260} - \frac{\langle xyxyyyy\rangle}{2016} - \frac{\langle xyyxyyy\rangle}{5040} + \frac{\langle xyyyyyy\rangle}{30\,240},\\
&\frac{\langle xxxxxxyy\rangle}{60\,480} - \frac{\langle xxxxxyxy\rangle}{15\,120} - \frac{\langle xxxxxyyy\rangle}{10\,080} + \frac{\langle xxxxyxxy\rangle}{20\,160} - \frac{\langle xxxxyxyy\rangle}{20\,160} + \frac{\langle xxxxyyxy\rangle}{2520} +\\
&\frac{23\,\langle xxxxyyyy\rangle}{120\,960} + \frac{\langle xxxyxxyy\rangle}{4032} - \frac{\langle xxxyxyxy\rangle}{10\,080} + \frac{13\,\langle xxxyxyyy\rangle}{30\,240} + \frac{\langle xxxyyxyy\rangle}{20\,160} -\\
&\frac{\langle xxxyyyxy\rangle}{3024} - \frac{\langle xxxyyyyy\rangle}{10\,080} + \frac{\langle xxyxyxyy\rangle}{2520} - \frac{\langle xxyxyyyy\rangle}{4032} - \frac{\langle xxyyxyyy\rangle}{10\,080} + \frac{\langle xxyyyyyy\rangle}{60\,480}\Big]
\end{aligned}$$

```
ApplyMorphism[mor_, w_LW] /; Deg[w] == 1 := w /. mor;
ApplyMorphism[mor_, w_LW] :=
  b @@ (ApplyMorphism[mor , #] & /@ LyndonFactorization[w]);
ApplyMorphism[mor_, s_LieSeries] := Module[
   {l = Length[s]},
   MakeLieSeries[l, Sum[
     ApplyMorphism[
      MakeLieSeries[l - k + 1, mor],
      s[[k]]
     ],
     {k, l}
    ]]
  ];
ApplyMorphism[mor_, expr_] := Expand[expr /. w_LW :→ ApplyMorphism[mor, w]];
BCH[n_, x_, y_] := ApplyMorphism[{LW["x"] → x, LW["y"] → y}, BCH[n]];
```

```
MakeLieSeries[3, {LW["x"] → LW["x"], LW["y"] → LW["z"]}]
```

$\{\langle x\rangle \to \text{LieSeries}[\langle x\rangle, 0, 0], \langle y\rangle \to \text{LieSeries}[\langle z\rangle, 0, 0]\}$

```
MakeLieSeries[2, LW["x"]]
```

$\text{LieSeries}[\langle x\rangle, 0]$

```
BCH[3]
```

$$\text{LieSeries}\left[\langle x\rangle + \langle y\rangle, \frac{\langle xy\rangle}{2}, \frac{\langle xxy\rangle}{12} + \frac{\langle xyy\rangle}{12}\right]$$

```
BCH[3, LW["y"], LW["z"]]
```

$$\text{LieSeries}\left[\langle y\rangle + \langle z\rangle, \frac{\langle yz\rangle}{2}, \frac{\langle yyz\rangle}{12} + \frac{\langle yzz\rangle}{12}\right]$$

```
BCH[3, LW["x"], BCH[3, LW["y"], LW["z"]]]
```

$$\text{LieSeries}\left[\langle x\rangle + \langle y\rangle + \langle z\rangle, \frac{\langle xy\rangle}{2} + \frac{\langle xz\rangle}{2} + \frac{\langle yz\rangle}{2}, \frac{\langle xxy\rangle}{12} + \frac{\langle xxz\rangle}{12} + \frac{\langle xyy\rangle}{12} + \frac{\langle xyz\rangle}{3} + \frac{\langle xzy\rangle}{6} + \frac{\langle xzz\rangle}{12} + \frac{\langle yyz\rangle}{12} + \frac{\langle yzz\rangle}{12}\right]$$

```
BCH[3, BCH[3, LW["x"], LW["y"]], LW["z"]]
```

$$\text{LieSeries}\left[\langle x\rangle + \langle y\rangle + \langle z\rangle, \frac{\langle xy\rangle}{2} + \frac{\langle xz\rangle}{2} + \frac{\langle yz\rangle}{2}, \frac{\langle xxy\rangle}{12} + \frac{\langle xxz\rangle}{12} + \frac{\langle xyy\rangle}{12} + \frac{\langle xyz\rangle}{3} + \frac{\langle xzy\rangle}{6} + \frac{\langle xzz\rangle}{12} + \frac{\langle yyz\rangle}{12} + \frac{\langle yzz\rangle}{12}\right]$$

```
{n = 4,
 t1 = BCH[n, LW["x"], BCH[n, LW["y"], LW["z"]]],
 t2 = BCH[n, BCH[n, LW["x"], LW["y"]], LW["z"]],
 t1 == t2}
```

$$\Big\{4,\ \text{LieSeries}\Big[\langle x\rangle+\langle y\rangle+\langle z\rangle,\ \frac{\langle xy\rangle}{2}+\frac{\langle xz\rangle}{2}+\frac{\langle yz\rangle}{2},\ \frac{\langle xxy\rangle}{12}+\frac{\langle xxz\rangle}{12}+\frac{\langle xyy\rangle}{12}+\frac{\langle xyz\rangle}{3}+\frac{\langle xzy\rangle}{6}+\frac{\langle xzz\rangle}{12}+\frac{\langle yyz\rangle}{12}+\frac{\langle yzz\rangle}{12},\ \frac{\langle xxyy\rangle}{24}+\frac{\langle xxyz\rangle}{12}+\frac{\langle xxzy\rangle}{12}+\frac{\langle xxzz\rangle}{24}+\frac{\langle xyyz\rangle}{12}+\frac{\langle xyzy\rangle}{12}+\frac{\langle xyzz\rangle}{12}+\frac{\langle xzyz\rangle}{12}+\frac{\langle yyzz\rangle}{24}\Big],$$
$$\text{LieSeries}\Big[\langle x\rangle+\langle y\rangle+\langle z\rangle,\ \frac{\langle xy\rangle}{2}+\frac{\langle xz\rangle}{2}+\frac{\langle yz\rangle}{2},\ \frac{\langle xxy\rangle}{12}+\frac{\langle xxz\rangle}{12}+\frac{\langle xyy\rangle}{12}+\frac{\langle xyz\rangle}{3}+\frac{\langle xzy\rangle}{6}+\frac{\langle xzz\rangle}{12}+\frac{\langle yyz\rangle}{12}+\frac{\langle yzz\rangle}{12},\ \frac{\langle xxyy\rangle}{24}+\frac{\langle xxyz\rangle}{12}+\frac{\langle xxzy\rangle}{12}+\frac{\langle xxzz\rangle}{24}+\frac{\langle xyyz\rangle}{12}+\frac{\langle xyzy\rangle}{12}+\frac{\langle xyzz\rangle}{12}+\frac{\langle xzyz\rangle}{12}+\frac{\langle yyzz\rangle}{24}\Big],\ \text{True}\Big\}$$

```
BCH[6, LW["y"], LW["z"]]
```

$$\text{LieSeries}\Big[\langle y\rangle+\langle z\rangle,\ \frac{\langle yz\rangle}{2},\ \frac{\langle yyz\rangle}{12}+\frac{\langle yzz\rangle}{12},\ \frac{\langle yyzz\rangle}{24},\ -\frac{\langle yyyyz\rangle}{720}+\frac{\langle yyyzz\rangle}{180}+\frac{\langle yyzyz\rangle}{360}+\frac{\langle yyzzz\rangle}{180}+\frac{\langle yzyzz\rangle}{120}-\frac{\langle yzzzz\rangle}{720},\ -\frac{\langle yyyyzz\rangle}{1440}+\frac{\langle yyyzyz\rangle}{720}+\frac{\langle yyyzzz\rangle}{360}+\frac{\langle yyzyzz\rangle}{240}-\frac{\langle yyzzzz\rangle}{1440}\Big]$$

```
BCH[6, LW["x"], BCH[6, LW["y"], LW["z"]]]
```

$$\text{LieSeries}\Big[\langle x\rangle+\langle y\rangle+\langle z\rangle,\ \frac{\langle xy\rangle}{2}+\frac{\langle xz\rangle}{2}+\frac{\langle yz\rangle}{2},$$

$$\frac{\langle xxy\rangle}{12}+\frac{\langle xxz\rangle}{12}+\frac{\langle xyy\rangle}{12}+\frac{\langle xyz\rangle}{3}+\frac{\langle xzy\rangle}{6}+\frac{\langle xzz\rangle}{12}+\frac{\langle yyz\rangle}{12}+\frac{\langle yzz\rangle}{12},$$

$$\frac{\langle xxyy\rangle}{24}+\frac{\langle xxyz\rangle}{12}+\frac{\langle xxzy\rangle}{12}+\frac{\langle xxzz\rangle}{24}+\frac{\langle xyyz\rangle}{12}+\frac{\langle xyzy\rangle}{12}+\frac{\langle xyzz\rangle}{12}+\frac{\langle xzyz\rangle}{12}+\frac{\langle yyzz\rangle}{24},$$

$$-\frac{\langle xxxxy\rangle}{720}-\frac{\langle xxxxz\rangle}{720}+\frac{\langle xxxyy\rangle}{180}+\frac{\langle xxxyz\rangle}{180}+\frac{\langle xxxzy\rangle}{90}+\frac{\langle xxxzz\rangle}{180}+\frac{\langle xxyxy\rangle}{360}+\frac{\langle xxyxz\rangle}{360}+$$
$$\frac{\langle xxyyy\rangle}{180}+\frac{\langle xxyyz\rangle}{30}+\frac{7\langle xxyzy\rangle}{120}+\frac{\langle xxyzz\rangle}{30}+\frac{\langle xxzxy\rangle}{180}+\frac{\langle xxzxz\rangle}{360}+\frac{\langle xxzyy\rangle}{60}+\frac{7\langle xxzyz\rangle}{120}+$$
$$\frac{\langle xxzzy\rangle}{60}+\frac{\langle xxzzz\rangle}{180}+\frac{\langle xyxyy\rangle}{120}+\frac{\langle xyxyz\rangle}{120}+\frac{\langle xyxzy\rangle}{60}+\frac{\langle xyxzz\rangle}{120}-\frac{\langle xyyxz\rangle}{120}-\frac{\langle xyyyy\rangle}{720}+$$
$$\frac{\langle xyyyz\rangle}{180}+\frac{\langle xyyzy\rangle}{120}+\frac{\langle xyyzz\rangle}{30}-\frac{\langle xyzxz\rangle}{120}-\frac{\langle xyzyy\rangle}{120}+\frac{\langle xyzyz\rangle}{60}+\frac{\langle xyzzy\rangle}{120}+\frac{\langle xyzzz\rangle}{180}+$$
$$\frac{\langle xzxzy\rangle}{60}+\frac{\langle xzxzz\rangle}{120}-\frac{\langle xzyyy\rangle}{180}+\frac{\langle xzyyz\rangle}{120}-\frac{\langle xzyzy\rangle}{60}+\frac{\langle xzyzz\rangle}{120}-\frac{\langle xzzyy\rangle}{120}-\frac{\langle xzzyz\rangle}{120}-$$
$$\frac{\langle xzzzy\rangle}{180}-\frac{\langle xzzzz\rangle}{720}-\frac{\langle yyyyz\rangle}{720}+\frac{\langle yyyzz\rangle}{180}+\frac{\langle yyzyz\rangle}{360}+\frac{\langle yyzzz\rangle}{180}+\frac{\langle yzyzz\rangle}{120}-\frac{\langle yzzzz\rangle}{720},$$

$$-\frac{\langle xxxxyy\rangle}{1440}-\frac{\langle xxxxyz\rangle}{720}-\frac{\langle xxxxzy\rangle}{720}-\frac{\langle xxxxzz\rangle}{1440}+\frac{\langle xxxyxy\rangle}{720}+\frac{\langle xxxyxz\rangle}{720}+\frac{\langle xxxyyy\rangle}{360}+$$
$$\frac{\langle xxxyyz\rangle}{180}+\frac{\langle xxxyzy\rangle}{72}+\frac{\langle xxxyzz\rangle}{180}+\frac{\langle xxxzxy\rangle}{360}+\frac{\langle xxxzxz\rangle}{720}+\frac{\langle xxxzyy\rangle}{120}+\frac{\langle xxxzyz\rangle}{72}+\frac{\langle xxxzzy\rangle}{120}+$$
$$\frac{\langle xxxzzz\rangle}{360}+\frac{\langle xxyxyy\rangle}{240}+\frac{\langle xxyxyz\rangle}{180}+\frac{\langle xxyxzy\rangle}{120}+\frac{\langle xxyxzz\rangle}{240}-\frac{\langle xxyyxz\rangle}{240}-\frac{\langle xxyyyy\rangle}{1440}+\frac{\langle xxyyyz\rangle}{180}+$$
$$\frac{\langle xxyyzy\rangle}{80}+\frac{\langle xxyyzz\rangle}{80}+\frac{\langle xxyzxy\rangle}{360}-\frac{\langle xxyzxz\rangle}{360}+\frac{\langle xxyzyy\rangle}{240}+\frac{\langle xxyzyz\rangle}{60}+\frac{\langle xxyzzy\rangle}{80}+\frac{\langle xxyzzz\rangle}{180}+$$
$$\frac{\langle xxzxyz\rangle}{360}+\frac{\langle xxzxzy\rangle}{120}+\frac{\langle xxzxzz\rangle}{240}-\frac{\langle xxzyyy\rangle}{360}+\frac{\langle xxzyyz\rangle}{80}+\frac{\langle xxzyzy\rangle}{120}+\frac{\langle xxzyzz\rangle}{80}-\frac{\langle xxzzyy\rangle}{240}+$$
$$\frac{\langle xxzzyz\rangle}{240}-\frac{\langle xxzzzy\rangle}{360}-\frac{\langle xxzzzz\rangle}{1440}+\frac{\langle xyxyyz\rangle}{240}+\frac{\langle xyxyzy\rangle}{120}+\frac{\langle xyxyzz\rangle}{240}+\frac{\langle xyxzyz\rangle}{120}-$$
$$\frac{\langle xyyxyz\rangle}{240}-\frac{\langle xyyyyz\rangle}{720}-\frac{\langle xyyyzy\rangle}{360}+\frac{\langle xyyyzz\rangle}{180}-\frac{\langle xyyzxz\rangle}{240}-\frac{\langle xyyzyy\rangle}{240}+\frac{\langle xyyzyz\rangle}{360}+$$
$$\frac{\langle xyyzzy\rangle}{240}+\frac{\langle xyyzzz\rangle}{180}+\frac{\langle xyzxzy\rangle}{120}+\frac{\langle xyzxzz\rangle}{240}-\frac{\langle xyzyxz\rangle}{120}-\frac{\langle xyzyyy\rangle}{360}-\frac{\langle xyzyzy\rangle}{120}+$$
$$\frac{\langle xyzyzz\rangle}{120}-\frac{\langle xyzzxz\rangle}{240}-\frac{\langle xyzzyy\rangle}{240}-\frac{\langle xyzzzy\rangle}{360}-\frac{\langle xyzzzz\rangle}{720}+\frac{\langle xzxzyz\rangle}{120}-\frac{\langle xzyyyz\rangle}{360}-$$
$$\frac{\langle xzyyzy\rangle}{120}+\frac{\langle xzyyzz\rangle}{240}-\frac{\langle xzyzyy\rangle}{120}-\frac{\langle xzyzyz\rangle}{120}-\frac{\langle xzyzzy\rangle}{120}-\frac{\langle xzyzzz\rangle}{360}-\frac{\langle xzzyyz\rangle}{240}-$$
$$\frac{\langle xzzyzy\rangle}{120}-\frac{\langle xzzyzz\rangle}{240}-\frac{\langle xzzzyz\rangle}{360}-\frac{\langle yyyyzz\rangle}{1440}+\frac{\langle yyyzyz\rangle}{720}+\frac{\langle yyyzzz\rangle}{360}+\frac{\langle yyzyzz\rangle}{240}-\frac{\langle yyzzzz\rangle}{1440}\Big]$$

```
Timing[{n = 6,
  Length /@ (t1 = BCH[n, LW["x"], BCH[n, LW["y"], LW["z"]]]),
  Length /@ (t2 = BCH[n, BCH[n, LW["x"], LW["y"]], LW["z"]]),
  t1 == t2}]
```

```
{0.094, {6, LieSeries[3, 3, 8, 9, 48, 82], LieSeries[3, 3, 8, 9, 48, 82], True}}
```

```
Timing[{n = 7,
  Timing[Length /@ (t1 = BCH[n, LW["x"], BCH[n, LW["y"], LW["z"]]])],
  Timing[Length /@ (t2 = BCH[n, BCH[n, LW["x"], LW["y"]], LW["z"]])],
  t1 == t2}]
```

```
{545.785, {7, {267.495, LieSeries[3, 3, 8, 9, 48, 82, 312]},
  {278.29, LieSeries[3, 3, 8, 9, 48, 82, 312]}, True}}
```